Contents

Abstract

The automated surveillance system employs an HC-SR04 ultrasonic sensor mounted on a servo-driven mechanism to scan a predefined area for intrusions. When an object crosses the configured distance threshold, an Arduino Nano issues a UART command to an ESP32-CAM module, which captures an image and transmits it via email using SMTP over Wi-Fi. This design delivers real-time visual alerts, offering a low-cost and easily deployable security solution for homes and offices. Its modular architecture allows flexible adjustment of detection range and notification settings.

The Project is based upon the 6 box model of embedded systems.



Figure 1: Six-Box Model of the Automated Surveillance System

Introduction

Intrusion detection systems often rely on high-resolution cameras and complex networking, driving up costs and complicating deployment in residential or small-business settings. A streamlined alternative using microcontrollers and simple sensors can provide accessible, real-time monitoring without significant infrastructure.

This report details an automated surveillance solution integrating an Arduino Nano, HC-SR04 ultrasonic sensor, SG90 servo motor, and ESP32-CAM module. The system continuously scans its environment, triggers image capture upon proximity detection, and emails snapshots to a registered address. We outline the hardware components, software flow, communication protocols, and implementation steps, followed by testing results and performance analysis.

1 System Architecture

1.1 Detailed Schematic of the Project

The schematic diagram shows all electrical connections between components. The Arduino Nano interfaces with the HC-SR04 sensor (trigger pin D2, echo pin D3) and controls the SG90 servo (PWM pin D9). The system includes a laser diode (D11) and buzzer (D10) for visual and audible alerts. UART communication occurs between Arduino TX (D1) and ESP32-CAM U0R with a voltage divider for level shifting.



Figure 2: Detailed Schematic of the Automated Surveillance System with Email alerts

2 Hardware Components

2.1 Arduino Nano (ATmega328)

The Arduino Nano serves as the primary controller, processing sensor data and making decisions. It features an ATmega328 microcontroller operating at 16MHz with 32KB flash memory, 2KB SRAM, and 1KB EEPROM. The board provides 14 digital I/O pins (6 with PWM) and 8 analog inputs, operating at 5V logic level. In our system, it handles servo control, distance measurement, and UART communication with the ESP32-CAM.



Figure 3: Arduino Nano Microcontroller Board

2.2 ESP32-CAM Module

The ESP32-CAM combines an ESP32-S chip with an RH-YX M21-45, 2-megapixel camera and microSD card slot. It features dual-core 32-bit CPU, 520KB SRAM, 4MB flash memory, and integrated Wi-Fi and Bluetooth connectivity. The module operates at 3.3V and includes a programmable LED flash. In our system, it handles image capture and email transmission over Wi-Fi.

Figure 4: ESP32-CAM Module with RH-YX M21-45 Camera

2.3 HC-SR04 Ultrasonic Sensor

The HC-SR04 ultrasonic sensor provides non-contact distance measurement with a range of 2cm to 400cm and accuracy of ±3mm. It operates at 5V and uses ultrasonic pulses (40kHz) to detect objects. The sensor has four pins: VCC, GND, Trigger (input), and Echo (output). In our system, it detects potential intruders by measuring their distance from the monitoring point.



Figure 5: HC-SR04 Ultrasonic Distance Sensor

2.4 SG90 Servo Motor

The SG90 is a lightweight micro servo motor with 180° rotation capability. It operates at 4.8-6V, drawing 220-250mA, and provides 1.8kg-cm of torque. The servo has three wires: power (red), ground (brown), and control signal (yellow). In our system, it enables the ultrasonic sensor to scan across a wide area by rotating it through a predefined arc.



Figure 6: SG90 Micro Servo Motor Schematic



Figure 7: SG90 Micro Servo Motor

2.5 650 nm Laser Diode

The 650nm red laser diode module provides a visible indicator when an intrusion is detected. It operates at 5V with a current draw of approximately 20mA and is connected to pin D11 through

a 220 ohm resistor. The module includes a built-in resistor for direct connection to the Arduino. In our system, it serves as a visual deterrent and indicator of system activation.



Figure 8: 650nm Red Laser Diode Module

2.6 Buzzer

The piezoelectric buzzer provides an audible alert when intrusion is detected. It operates at 3-6V with approximately 30mA current draw and produces a sound of about 85dB. Connected to pin D10 of the Arduino Nano, it complements the visual alert from the laser diode to create a multi-sensory alarm.

Figure 9: Piezoelectric Buzzer

2.7 Power Supply & Level Shifting

The system requires two voltage levels: 5V for the Arduino Nano, sensors, and servo; and 3.3V for the ESP32-CAM. A 5V external supply is provided, while the ESP32-CAM has an onboard 3.3V regulator. For UART communication between the Arduino (5V logic) and ESP32-CAM

(3.3V logic), a voltage divider using 1.66k and 3.3k resistors converts the 5V TX signal from Arduino to a safe 3.3V level for the ESP32-CAM U0R pin. This prevents damage to the ESP32-CAM while ensuring reliable communication.

3 Software Design

3.1 Overview & Flowchart

The software architecture consists of two main components: the Arduino Nano firmware and the ESP32-CAM firmware. The Arduino continuously sweeps the servo motor through a 180° arc while measuring distances using the ultrasonic sensor. When an object is detected within the threshold distance, the Arduino activates the laser and buzzer alerts and sends a ”CAPTURE” command via UART to the ESP32-CAM. The ESP32-CAM then captures an image and sends it via email using the configured SMTP server.



Figure 10: Integrated System Flowchart

3.2 Arduino Nano Sketch

3.2.1 Sensor Scanning Loop

The Arduino controls the SG90 servo to sweep from 0° to 180° and back in 1° increments. At each position, the HC-SR04 sensor measures the distance by sending a 10s pulse on the trigger pin (D2) and measuring the duration of the echo pulse on pin D3. The distance is calculated using the formula: distance = duration / 58. The system pauses briefly (15ms) at each position to ensure stable readings and prevent mechanical jitter.

3.2.2 Threshold Check & Alerts

When an object is detected within the predefined threshold distance (default: 20cm), the system stops the servo motion to prevent motion blur, activates the laser diode (D11) and buzzer (D10) for visual and audible alerts. The alerts remain active until the object is removed from the detection zone. The system implements a state-based approach to track object presence and avoid rapid triggering.

3.2.3 UART Command Transmission

Upon detection, the Arduino sends the ASCII string ”CAPTURE” to the ESP32-CAM via UART at 115200 baud rate (8 data bits, no parity, 1 stop bit). The system implements a one-way communication protocol where the Arduino sends commands without expecting acknowledgment from the ESP32-CAM. After sending the command, the system waits for 500ms to avoid immediate re-triggering.

3.3 ESP32-CAM Sketch

3.3.1 UART Reception

The ESP32-CAM continuously monitors its UART interface for incoming commands. When a line exactly equal to ”CAPTURE” is received, it proceeds with image capture. The sketch implements a command parsing approach that reads complete lines from Serial and trims whitespace to ensure reliable command recognition.

3.3.2 Image Capture & Storage

The ESP32-CAM uses the esp_camera.h library to initialize the camera with FRAMESIZE_SVGA resolution.Upon receiving the capture command, it acquires a frame in RGB565 format, converts it to JPEG with 80% quality, and prepares it for email transmission. The camera is configured

with the AI_THINKER model settings, which is appropriate for the ESP32-CAM module being used.

3.3.3 SMTP Email Configuration

The ESP32-CAM uses the ESP_Mail_Client library to handle secure email transmission. The sketch configures SMTP with the following settings:

- Server: smtp.gmail.com
- Port: 587 (STARTTLS)
- Authentication: Gmail account with app password
- Email Content: HTML with inline image attachment
- Image Format: JPEG with Base64 encoding

The email includes a simple HTML body with the camera image embedded inline. Debug outputs are minimized to avoid interference with the UART communication channel.

4 Communication Protocols

4.1 UART

UART (Universal Asynchronous Receiver-Transmitter) provides the communication link between the Arduino Nano and ESP32-CAM:

- Baud Rate: 115200 bps (bits per second)
- Format: 8 data bits, no parity, 1 stop bit (8N1)
- Command: ASCII string ”CAPTURE” (8 bytes including newline)
- Direction: One-way communication (Arduino Nano to ESP32-CAM)
- Connection: Arduino Nano TX (D1) $\rightarrow$ Voltage Divider $\rightarrow$ ESP32-CAM U0R

4.2 SMTP over TLS

SMTP (Simple Mail Transfer Protocol) with TLS (Transport Layer Security) provides secure email transmission:

- Server: smtp.gmail.com

- Port: 587 (STARTTLS)
- Authentication: Gmail account with app password
- Security: TLS 1.2 or higher
- Message Format: HTML with inline image
- Image Format: JPEG with Base64 encoding

5 Implementation Details

5.1 Wiring Connections Table

The following table details all electrical connections between components:

Table 1: Wiring Connection Table for Automated Surveillance System with Email Alerts

Component	**Pin**	**Connected To**	**Notes**
HC-SR04	VCC	Arduino 5V	Power supply
HC-SR04	GND	Arduino GND	Ground connection
HC-SR04	Trig	Arduino D2	Trigger pulse output
HC-SR04	Echo	Arduino D3	Echo pulse input
SG90 Servo	Red	Arduino 5V	Power supply
SG90 Servo	Brown	Arduino GND	Ground connection
SG90 Servo	Yellow	Arduino D9	PWM control signal
Laser Diode	VCC	Arduino D11	Through 220 ohm resistor
Laser Diode	GND	Arduino GND	Ground connection
Buzzer	+	Arduino D10	Digital control
Buzzer	-	Arduino GND	Ground connection
Arduino Nano	TX (D1)	Voltage Divider	Serial data output
Voltage Divider	Output	ESP32-CAM U0R	Level-shifted to 3.3V
ESP32-CAM	5V	Power Supply 5V	Main power input
ESP32-CAM	GND	Power Supply GND	Ground connection
ESP32-CAM	U0R	Arduino TX via divider	UART receive

5.2 Power Budget

The system's power requirements are detailed below:

Table 2: Power Consumption Budget

Component	**Current (mA)**	**Voltage (V)**	**Notes**
Arduino Nano	19	5.0	Active mode
ESP32-CAM	300-400	3.3	WiFi active, camera on
HC-SR04	15	5.0	During measurement
SG90 Servo	150-250	5.0	During movement (peak: 250mA)
Laser Diode	20	5.0	When activated
Buzzer	30	5.0	When activated
Total (Idle)	530	5.0	No alerts active
Total (Peak)	730	5.0	All components active

With a maximum current draw of approximately 2A at 5V during peak operation (including transient current spikes during WiFi transmission and servo movement), a standard 5V/2A power supply or USB power bank provides sufficient power with adequate margin for stable operation.

5.3 Prototype Assembly

The prototype was assembled on a standard green zero PCB board using direct soldering. The Arduino Nano was mounted centrally with stable power supply of 5V at the corner ,the ESP32-CAM mounted on the servo motor. The servo motor was attached to the board's edge using mounting brackets, with the ultrasonic sensor secured to the servo horn using hot glue and double sided tapes.



Figure 11: Complete assembled prototype



Figure 12: Prototype Assembly on Green Zero PCB

All connections were made with 24AWG solid-core wire, with heat-shrink tubing used at junctions to prevent short circuits. The voltage divider for UART level shifting was implemented directly on the board using 1.66k and 3.3k resistors. For the laser diode, a 220 ohm current-limiting resistor was used to protect the LED.

6 Testing Validation

6.1 Functional Tests

The system underwent comprehensive functional testing to ensure reliable operation in various conditions:

- **Detection Accuracy**: Objects of various sizes (10cm, 20cm, 30cm diameter) were placed at different distances (15cm to 200cm) and angles relative to the sensor. The system achieved excellent detection reliability for objects larger than 20cm at distances up to 150cm. The servo-mounted ultrasonic sensor provided consistent readings throughout its 180° sweep, ensuring comprehensive coverage of the monitored area.

- **Email Latency**: The time from detection to email receipt was measured across 50 trials. The average latency was 4.2 seconds (±1.1s), with 90% of emails delivered within 5 seconds. This rapid notification ensures that users can respond quickly to potential security threats. The system's use of port 587 with STARTTLS provided reliable email delivery even through restrictive network environments.

- **Battery Life**: Using a 12V,2200mAh(common Li-ion pack) battery, the system operated continuously for approximately 6 hours before requiring recharging. This duration makes it suitable for temporary deployments or as a backup during power outages. The system's power management features, including sleep modes for the ESP32-CAM when not actively transmitting, helped extend battery life.

6.2 Edge Case Handling

The system was tested under various adverse conditions to evaluate its robustness:

- **Power Interruption**: After sudden power loss and restoration, both the Arduino Nano and ESP32-CAM recovered automatically and resumed normal operation without manual intervention. This resilience is critical for a security system that must operate reliably without constant supervision. The system's configuration parameters are stored in non-volatile memory to preserve settings across power cycles.

- **Rapid Triggers**: When multiple intrusions occurred in rapid succession, the system implemented a cooldown period (configurable, default: 10 seconds) to prevent email flooding while maintaining detection capability. This balanced approach ensures users receive timely alerts without overwhelming their inbox during continuous motion events.

- **Low Light Conditions**: The ESP32-CAM's built-in LED flash was activated during image capture in low-light environments, ensuring usable images even in near-darkness. This feature significantly enhances the system's effectiveness as a 24-hour surveillance solution, providing clear visual confirmation regardless of ambient lighting conditions.

7 Results & Discussion

The automated surveillance system demonstrated reliable performance in detecting intrusions and delivering timely email alerts with visual confirmation. Key performance metrics include:

- Detection range: 10cm to 150cm (reliable), up to 400cm (with reduced reliability)
- Detection angle: 180° horizontal sweep
- Image resolution: SVGA (800×600 pixels)
- Alert latency: 4.2 seconds average from detection to email delivery
- Power consumption: 214mA (idle), up to 2A (peak) at 5V
- Operational duration: Approximately 8 hours on 2200mAh battery

The system's primary limitations include:

- **Limited Field of View**: The camera's fixed position captured only the area directly in front of the device, potentially missing activity outside its field of view even if detected by the ultrasonic sensor.
- **Environmental Sensitivity**: The ultrasonic sensor's performance degraded in environments with soft or sound-absorbing surfaces, potentially missing some intrusion events in heavily furnished spaces.
- **Wi-Fi Dependency**: Alert delivery relied on stable Wi-Fi connectivity, though local storage mitigated temporary outages. In locations with poor Wi-Fi coverage, the system's effectiveness as a real-time alert mechanism was reduced.
- **Power Requirements**: Continuous operation required external power or frequent battery replacement, limiting deployment options in locations without accessible power sources.

Future improvements could include:

- Adding PIR (Passive Infrared) sensors to complement ultrasonic detection
- Implementing a pan-tilt mechanism for the camera to track detected objects

- Adding cellular connectivity as a backup to Wi-Fi
- Implementing AI-based image analysis to reduce false positives
- Developing a mobile application for remote monitoring and configuration

The project successfully demonstrated that affordable microcontrollers and sensors can create an effective surveillance system suitable for residential and small business applications. The modular design allows for easy customization and expansion to meet specific security requirements.

References

1. Arduino Nano Technical Specifications, https://store.arduino.cc/products/arduino-nano
2. ESP32-CAM AI-Thinker Datasheet, https://ai-thinker.com/product/esp32-cam/
3. HC-SR04 Ultrasonic Sensor Datasheet, https://cdn.sparkfun.com/datasheets/Sensors/Proximity/HCSR04.pdf
4. SG90 Micro Servo Datasheet, http://www.ee.ic.ac.uk/pcheung/teaching/DE1_EE/stores/sg90_datasheet.pdf
5. ESP32 Arduino Core Documentation, https://github.com/espressif/arduino-esp32
6. ESP_Mail_Client Library, https://github.com/mobizt/ESP-Mail-Client
7. ESP32 Camera Library, https://github.com/espressif/esp32-camera

Appendices

Appendix A: Arduino Nano Code

```
/**
 * Nano Board Code to Trigger ESP32CAM Capture via UART Command
 *
 * When an object is detected within a threshold distance, the servo motion stops
 * (preventing motion blur) and the Nano sends a "CAPTURE" command via Serial.
 * The ESP32CAM board (with its own code) will listen for this command to capture
 * a photo and send it via email.
```

```
 */

#include <Arduino.h>
#include <Servo.h>

// ----- Pin Definitions -----
const int trigPin = 2;          // Ultrasonic sensor TRIG pin
const int echoPin = 3;          // Ultrasonic sensor ECHO pin
const int servoPin = 9;         // Servo control pin (SG90)
const int buzzerPin = 10;       // Buzzer pin (via transistor)
const int laserPin = 11;        // Laser LED pin (with resistor)
// Note: No dedicated ESP32CAM trigger pin is used here.


// ----- Other Parameters -----
const int thresholdDistance = 20; // Distance in cm below which an object
    is considered "detected"

// ----- Servo and state variables -----
Servo servoMotor;
int servoAngle = 0;
int direction = 1;           // 1: increasing angle, -1: decreasing angle
bool objectDetected = false;

// Function to read distance from an HC-SR04 ultrasonic sensor.
long readUltrasonicDistance() {
  // Ensure TRIG is LOW initially.
  digitalWrite(trigPin, LOW);
  delayMicroseconds(2);

  // Send a 10-microsecond pulse.
  digitalWrite(trigPin, HIGH);
  delayMicroseconds(10);
  digitalWrite(trigPin, LOW);

  // Read the pulse duration from echoPin; timeout after 30000 s  (30ms).
  long duration = pulseIn(echoPin, HIGH, 30000);
  long distance = duration / 58;  // Convert duration to centimeters.
  return distance;
}

void setup() {
  Serial.begin(115200);

  // Set sensor pins.
```

```
  pinMode(trigPin, OUTPUT);
  pinMode(echoPin, INPUT);

  // Configure output pins.
  pinMode(buzzerPin, OUTPUT);
  pinMode(laserPin, OUTPUT);

  // Default states: buzzer off, laser off.
  digitalWrite(buzzerPin, LOW);
  digitalWrite(laserPin, LOW);

  // Setup and start the servo.
  servoMotor.attach(servoPin);
  servoMotor.write(servoAngle);
}

void loop() {
  long distance = readUltrasonicDistance();
  Serial.print("Distance: ");
  Serial.print(distance);
  Serial.println(" cm");

  // If an object is detected within the threshold distance.
  if (distance > 0 && distance < thresholdDistance) {
    if (!objectDetected) {
      objectDetected = true;
      Serial.println("Object detected. Stopping servo, activating alerts,
    and sending CAPTURE command.");

      // Stop the servo motion by not updating its position.
      // Activate buzzer and laser LED.
      digitalWrite(buzzerPin, HIGH);
      digitalWrite(laserPin, HIGH);

      // Send the capture command over Serial.
      // (Ensure that the command is sent on its own line for the
    ESP32CAM to recognize it.)
      Serial.println("CAPTURE");

      // Optionally, additional debug lines may appear before or after,
      // but the ESP32CAM code checks each complete line so only an
    exact "CAPTURE"
      // command will trigger a capture.

      // Delay to avoid immediate re-triggering.
```

```
      delay(500);
    }
  } else {
    if (objectDetected) {
      // Object is now removed. Reset state and deactivate alerts.
      objectDetected = false;
      digitalWrite(buzzerPin, LOW);
      digitalWrite(laserPin, LOW);
      Serial.println("Object removed. Resuming servo motion.");
    }

    // Continuous servo sweeping when no object is present.
    servoAngle += direction;
    if (servoAngle <= 0 || servoAngle >= 180) {
      direction = -direction;
    }
    servoMotor.write(servoAngle);
    delay(15);  // Adjust for smooth servo motion.
  }

  delay(50);  // Main loop delay.
}
```

Listing 1: Arduino Nano Sketch for Automated Surveillance System

Appendix B: ESP32-CAM Code

```
/**
 * Minimal Debug Version of ESP32-CAM Code
 * The ESP32-CAM waits for a UART command from the Arduino Nano.
 * When a line exactly equal to "CAPTURE" is received, it captures an image and sends it via email.
 * Debug outputs are minimized to avoid interference.
 */

#include <Arduino.h>
#if defined(ESP32)
  #include <WiFi.h>
#endif
#include <ESP_Mail_Client.h>
#include "esp_camera.h"

// ===================
// Select camera model
```

```
// ====================
#define CAMERA_MODEL_AI_THINKER // Has PSRAM
#include "camera_pins.h"

// WiFi credentials
#define WIFI_SSID "WIFI-SSID"
#define WIFI_PASSWORD "PASSWORD"

// SMTP settings
#define SMTP_HOST "smtp.gmail.com"
#define SMTP_PORT esp_mail_smtp_port_587

#define AUTHOR_EMAIL "mailzubair.2020@gmail.com"
#define AUTHOR_PASSWORD "<12 digit App-specific-password>"
#define RECIPIENT_EMAIL "mdzubairqureshi6@gmail.com"

// Uncomment to enable debug prints (if needed)
// #define DEBUG

#ifdef DEBUG
  #define DEBUG_PRINT(x) Serial.println(x)
#else
  #define DEBUG_PRINT(x) // Debug prints disabled
#endif

// Global SMTP session object.
SMTPSession smtp;

// Minimal SMTP callback.
void smtpCallback(SMTP_Status status) {
  // Keep empty to avoid extra output.
}

/**
 * @brief Captures an image from the camera, converts it to JPEG, and sends
    it via email.
 */
void captureAndSendEmail() {
  DEBUG_PRINT("Capture command received.");

  // Capture image from the camera.
  camera_fb_t *fb = esp_camera_fb_get();
  if (!fb) {
    return;
  }
```

```
61
62  // Ensure image is in the expected format; if not, return.
63  if (fb->format != PIXFORMAT_RGB565) {
64    esp_camera_fb_return(fb);
65    return;
66  }
67
68  // Convert the image from RGB565 to JPEG.
69  uint8_t *jpg_buf = NULL;
70  size_t jpg_size = 0;
71  if (!frame2jpg(fb, 80, &jpg_buf, &jpg_size)) {
72    esp_camera_fb_return(fb);
73    return;
74  }
75
76  // Prepare the email message with the captured image.
77  SMTP_Attachment att;
78  SMTP_Message message;
79
80  message.enable.chunking = true;
81  message.sender.name = F("ESP Mail");
82  message.sender.email = AUTHOR_EMAIL;
83  message.subject = F("Camera Image Capture");
84  message.addRecipient(F("user1"), RECIPIENT_EMAIL);
85  message.html.content = F("<span style=\"color:#ff0000;\">Camera image:</
     span><br/><br/><img src=\"cid:image-001\" alt=\"esp32 cam image\">");
86  message.html.transfer_encoding = Content_Transfer_Encoding::enc_7bit;
87  message.html.charSet = F("utf-8");
88
89  att.descr.filename = F("camera.jpg");
90  att.descr.mime = F("image/jpeg");
91  att.blob.data = jpg_buf;
92  att.blob.size = jpg_size;
93  att.descr.content_id = F("image-001");
94  att.descr.transfer_encoding = Content_Transfer_Encoding::enc_base64;
95  message.addInlineImage(att);
96
97  // Setup SMTP configuration.
98  Session_Config config;
99  config.server.host_name = SMTP_HOST;
100 config.server.port = SMTP_PORT;
101 config.login.email = AUTHOR_EMAIL;
102 config.login.password = AUTHOR_PASSWORD;
103 config.login.user_domain = F("127.0.0.1");
104 config.time.ntp_server = F("pool.ntp.org,time.nist.gov");
```

```
  config.time.gmt_offset = 3;
  config.time.day_light_offset = 0;

  if (!smtp.connect(&config)) {
    esp_camera_fb_return(fb);
    if (jpg_buf) free(jpg_buf);
    return;
  }

  if (!smtp.isLoggedIn()) {
    // No verbose output.
  }

  // Send the email.
  MailClient.sendMail(&smtp, &message, true);

  // Cleanup.
  esp_camera_fb_return(fb);
  if (jpg_buf) free(jpg_buf);
}

/**
 * Setup: initializes the camera, WiFi, and SMTP configuration.
 */
void setup() {
  Serial.begin(115200);
  // Optionally print startup only if debugging.
  DEBUG_PRINT("ESP32-CAM started.");

  // Configure the camera.
  camera_config_t camCfg;
  camCfg.ledc_channel = LEDC_CHANNEL_0;
  camCfg.ledc_timer = LEDC_TIMER_0;
  camCfg.pin_d0 = Y2_GPIO_NUM;
  camCfg.pin_d1 = Y3_GPIO_NUM;
  camCfg.pin_d2 = Y4_GPIO_NUM;
  camCfg.pin_d3 = Y5_GPIO_NUM;
  camCfg.pin_d4 = Y6_GPIO_NUM;
  camCfg.pin_d5 = Y7_GPIO_NUM;
  camCfg.pin_d6 = Y8_GPIO_NUM;
  camCfg.pin_d7 = Y9_GPIO_NUM;
  camCfg.pin_xclk = XCLK_GPIO_NUM;
  camCfg.pin_pclk = PCLK_GPIO_NUM;
  camCfg.pin_vsync = VSYNC_GPIO_NUM;
  camCfg.pin_href = HREF_GPIO_NUM;
```

```
  camCfg.pin_sscb_sda = SIOD_GPIO_NUM;
  camCfg.pin_sscb_scl = SIOC_GPIO_NUM;
  camCfg.pin_pwdn = PWDN_GPIO_NUM;
  camCfg.pin_reset = RESET_GPIO_NUM;
  camCfg.xclk_freq_hz = 8000000;
  camCfg.pixel_format = PIXFORMAT_RGB565;
  camCfg.frame_size = FRAMESIZE_SVGA;
  camCfg.jpeg_quality = 12;   // Do not change this.
  camCfg.fb_count = 1;

  if (esp_camera_init(&camCfg) != ESP_OK) {
    return;
  }

  // Connect to WiFi.
  WiFi.begin(WIFI_SSID, WIFI_PASSWORD);
  while (WiFi.status() != WL_CONNECTED) {
    delay(200);
  }

  // Setup SMTP; disable extra debug.
  MailClient.networkReconnect(true);
  smtp.debug(0);
  smtp.callback(smtpCallback);
}

/**
 * Main loop: reads complete lines from Serial and triggers capture only if
    the line equals "CAPTURE".
 */
void loop() {
  if (Serial.available()) {
    // Read until a newline character.
    String command = Serial.readStringUntil('\n');
    command.trim();

    if (command.equals("CAPTURE")) {
      captureAndSendEmail();
      delay(500);
    }
    // Ignore any other commands.
  }
  delay(10);
```

```
}
```

Listing 2: ESP32-CAM Sketch for Automated Surveillance System

Appendix C: Datasheets Library Links

- Arduino Servo Library: https://www.arduino.cc/reference/en/libraries/servo/
- NewPing Library for HC-SR04: https://github.com/microflo/NewPing
- ESP32 Camera Library: https://github.com/espressif/esp32-camera
- ESP Mail Client Library: https://github.com/mobizt/ESP-Mail-Client
- ESP32 Arduino Core: https://github.com/espressif/arduino-esp32
- Arduino Software Serial Library: https://www.arduino.cc/en/Reference/SoftwareSerial